Example 8.1 Wave Propagation in a Retangle
Example 8.1.1 is solved in Maple below:

```
> restart:with(inttrans):with(plots):
```

The governing equation is entered here:

```
>
> eq:=diff(u(x,t),t)+v*diff(u(x,t),x);
```

$$eq := \frac{\partial}{\partial t} u(x,t) + v \left(\frac{\partial}{\partial x} u(x,t) \right) \quad (1)$$

The initial and boundary conditions are entered here.

```
> u(x,0):=1;
```

$$u(x,0) := 1 \quad (2)$$

```
> bc1:=u(0,t)=0;
```

$$bc1 := u(0,t) = 0 \quad (3)$$

The governing equation and the boundary condition are converted to the Laplace domain:

```
> eqs:=laplace(eq,t,s);
```

$$eqs := s\, laplace(u(x,t), t, s) - 1 + v \left(\frac{\partial}{\partial x} laplace(u(x,t), t, s) \right) \quad (4)$$

```
> eqs:=subs(laplace(u(x,t),t,s)=U(x),eqs);
```

$$eqs := s\, U(x) - 1 + v \left(\frac{d}{dx} U(x) \right) \quad (5)$$

```
> bc1:=laplace(bc1,t,s);
```

$$bc1 := laplace(u(0,t), t, s) = 0 \quad (6)$$

```
> bc1:=subs(laplace(u(0,t),t,s)=U(0),bc1);
```

$$bc1 := U(0) = 0 \quad (7)$$

```
> U(x):=rhs(dsolve({eqs,bc1},U(x)));
```

$$U(x) := \frac{1}{s} - \frac{e^{-\frac{s x}{v}}}{s} \quad (8)$$

The solution obtained in the Laplace domain is converted to the time domain here:

```
> u:=invlaplace(U(x),s,t);
```

$$u := 1 - invlaplace\left(\frac{e^{-\frac{s x}{v}}}{s}, s, t \right) \quad (9)$$

The following plots can be obtained:

```
> plot3d(subs(v=1.,u),x=0..1,t=1e-6..1,axes=boxed,title="Figure
  Exp. 8.1.",labels=[x,t,"u"],orientation=[-137,50]);
```

Figure Exp. 8.1.




```
> plot([subs(v=1,t=0.1,u),subs(v=1,t=0.25,u),subs(v=1,t=0.5,u),
  subs(v=1,t=1,u)],x=0..1,axes=boxed,title="Figure Exp. 8.2.",
  thickness=5,labels=[x,"u"]);
```

Figure Exp. 8.2.




```
>
```